algol<
_b_e_g_i_n
   _r_e_a_l pi,EPS,EPS4;
   _i_n_t_e_g_e_r outputCode,oldrand;
    _p_r_o_c_e_d_u_r_e Pmult(A1, deg1, A0, deg0, a);
    _v_a_l_u_e deg0, a;
    _i_n_t_e_g_e_r deg1, deg0;
    _r_e_a_l a;
    _a_r_r_a_y A1, A0;
    _b_e_g_i_n
       _c_o_m_m_e_n_t P1(x) = P0(x)×(x-a);
       _i_n_t_e_g_e_r i;
       deg1 := deg0+1;
       _f_o_r i:=0 _s_t_e_p 1 _u_n_t_i_l deg0 _d_o
       A1[i+1]:=A0[i];
       A1[0]:=0;
       _f_o_r i:=0 _s_t_e_p 1 _u_n_t_i_l deg0 _d_o
       A1[i] := A1[i]-a×A0[i]
   _e_n_d Pmult;
   _r_e_a_l _p_r_o_c_e_d_u_r_e RANDOM;
   _c_o_d_e RANDOM,oldrand;
   1, 37;
   2, 44;
   pm a2,mln re1     ; M:=oldrand; RM:=oldrand×16807;
   dl re2,gm a2      ; M:=RM _m_o_d 2147483647;
   xr   , nkf 39     ; RF:=M;
   dkf re3, srf c44  ; RF:=M/2147483647.0-0.5;
   grf p-1, hv r4
e1: qq 16807.39
e2: qq 2147483647.39
_f
e3: 2147483647.0
    _e;
   _r_e_a_l _p_r_o_c_e_d_u_r_e CubeRoot(x);
   _v_a_l_u_e x;
   _r_e_a_l x;
   CubeRoot:=(_i_f x<0 _t_h_e_n
      -exp(ln(-x)/3)
      _e_l_s_e _i_f x>0 _t_h_e_n
      exp(ln(x)/3)
      _e_l_s_e
      0);
   _r_e_a_l _p_r_o_c_e_d_u_r_e atan2(y,x);
   _v_a_l_u_e y,x;
   _r_e_a_l y,x;
    atan2:=(
      _i_f x>0 _t_h_e_n
         arctan(y/x)
      _e_l_s_e _i_f x<0 _t_h_e_n
         arctan(y/x)+(_i_f y_>0 _t_h_e_n pi _e_l_s_e -pi)
      _e_l_s_e
         (_i_f y>0 _t_h_e_n pi/2 _e_l_s_e -pi/2));
   _p_r_o_c_e_d_u_r_e LinearRoot(a, z);
   _a_r_r_a_y a;
   _r_e_a_l z;
   _b_e_g_i_n
      _i_f a[2]=0 _t_h_e_n
        z:=0
      _e_l_s_e
        z:=-a[1]/a[2]
   _e_n_d LinearRoot;
   _p_r_o_c_e_d_u_r_e OneLargeTwoSmall(a1,a2,a4,w,z);
   _v_a_l_u_e a1,a2,a4,w;
   _r_e_a_l a1,a2,a4,w;

   _a_r_r_a_y z;
   _b_e_g_i_n
      _a_r_r_a_y aq[1:3];
      aq[1] := a1;
      aq[2] := a2+a1/w;
      aq[3] := -a4×w;
      QuadraticRoots(aq, z);
      z[3,1] := w;
      z[3,2] := 0;
      _i_f z[1,2] |=0 _t_h_e_n
      _b_e_g_i_n
         z[3,1]:=z[2,1]; z[3,2]:=z[2,2];
         z[2,1]:=z[1,1]; z[2,2]:=z[1,2];
         z[1,1]:=w;
         z[1,2]:=0
      _e_n_d
   _e_n_d OneLargeTwoSmall;
   _p_r_o_c_e_d_u_r_e QuadraticRoots(a, z);
   _a_r_r_a_y a,z;
   _b_e_g_i_n
      _r_e_a_l d, r, w, x, y;
      _i_f a[1]=0 _t_h_e_n
      _b_e_g_i_n
         z[1,1]:=0;
         z[1,2]:=0;
         z[2,1]:=-a[2]/a[3];
         z[2,2]:=0;
         outputCode:=21;
         _g_o_t_o end
      _e_n_d;
      d:=a[2]|∧2 - 4×a[1]×a[3];
      _i_f abs(d) _< 2×EPS×a[2]|∧2 _t_h_e_n
      _b_e_g_i_n
         z[1,1] := -0.5×a[2]/a[3];
         z[1,2] := 0;
         z[2,1] := z[1,1];
         z[2,2] := z[1,2];
         outputCode := 22;
         _g_o_t_o end
      _e_n_d;
      r := sqrt(abs(d));
      _i_f d<0 _t_h_e_n
      _b_e_g_i_n
         x := -0.5×a[2]/a[3];
         y := abs(0.5×r/a[3]);
         z[1,1] := x;
         z[1,2] := y;
         z[2,1] := x;
         z[2,2] := -y;
         outputCode := 23;
         _g_o_t_o end
      _e_n_d;
      _i_f a[2] |= 0 _t_h_e_n
      _b_e_g_i_n
         w := -(a[2]+sign(a[2])×abs(r));
         z[1,1] := 2×a[1]/w;
         z[1,2] := 0;
         z[2,1] := 0.5×w/a[3];
         z[2,2] := 0;
         outputCode := 22;
         _g_o_t_o end
      _e_n_d;
      x := abs(0.5×r/a[3]);
      z[1,1] := x;

      z[1,2] := 0;
      z[2,1] := -x;
      z[2,2] := 0;
      outputCode := 22;
end:
   _e_n_d QuadraticRoots;
   _p_r_o_c_e_d_u_r_e CubicRoots(a, z);
   _a_r_r_a_y a, z;
   _b_e_g_i_n
      _c_o_m_m_e_n_t
         PURPOSE - Compute the roots of the real polynomial
             a[1] + a[2]×z + ... + a[4]×z|∧3

         Note: It is assumed that a[4] is non-zero. No test is made here
      ;
      _r_e_a_l rt3,arg, c, cf, d, p, p1, q, q1, r, ra, rb, rq, rt, r1, s, sf, sq, sum, t, tol, t1, w, w1, w2, x, x1, x2, x3, y, y1, y2, y3;
      _a_r_r_a_y aq[1:3];
      rt3 := sqrt(3);
      _i_f a[1]=0 _t_h_e_n
      _b_e_g_i_n
         _c_o_m_m_e_n_t One root is obviously zero;
         _a_r_r_a_y a2[1:3], z2[1:2,1:2];
         a2[1] := a[2];
         a2[2] := a[3];
         a2[3] := a[4];
         _c_o_m_m_e_n_t remaining 2 roots here;
         QuadraticRoots(a2, z2);
         z[1,1] := 0;
         z[1,2] := 0;
         z[2,1] := z2[1,1];
         z[2,2] := z2[1,2];
         z[3,1] := z2[2,1];
         z[3,2] := z2[2,2];
         _g_o_t_o end
      _e_n_d;

      p:=a[3]/(3×a[4]);
      q:=a[2]/a[4];
      r:=a[1]/a[4];
      tol:=EPS4;

      c:=0;
      t:=a[2]-p×a[3];
      _i_f abs(t)>tol×abs(a[2]) _t_h_e_n c:=t/a[4];

      t:=2×p|∧2 - q;
      _i_f abs(t) _< tol×abs(q) _t_h_e_n t := 0;
      d := r + p×t;
      _i_f abs(t) _< tol×abs(r) _t_h_e_n _g_o_t_o l110;

      _c_o_m_m_e_n_t sq = (a[4]/s)|∧2 × (c|∧3/27 + d|∧2/4) ;

      s := abs(a[1]);
      _i_f abs(a[2])>s _t_h_e_n s:=abs(a[2]);
      _i_f abs(a[3])>s _t_h_e_n s:=abs(a[3]);
      p1 := a[3]/(3×s);
      q1 := a[2]/s;
      r1 := a[1]/s;

      t1 := q- 2.25×p|∧2;
      _i_f abs(t1) _< tol×abs(q) _t_h_e_n t1 := 0;
      w := 0.25×r1|∧2;
      w1 := 0.5×p1×r1×t;
      w2 := q1|∧2×t1/27;

      _i_f w1 _> 0 _t_h_e_n
      _b_e_g_i_n
         w := w+w1;
         sq := w+w2
      _e_n_d
      _e_l_s_e _i_f w2 < 0 _t_h_e_n
         sq := w + (w1+w2)
      _e_l_s_e
      _b_e_g_i_n
         w := w+w2;
         sq := w+w1
      _e_n_d;

      _i_f abs(sq) _< tol×w _t_h_e_n sq := 0;
      rq := abs(s/a[4])×sqrt(abs(sq));
      _i_f sq _> 0 _t_h_e_n _g_o_t_o l40;

      _c_o_m_m_e_n_t all roots are real;

      arg := atan2(rq, -0.5×d);
      cf := cos(arg/3);
      sf := sin(arg/3);
      rt := sqrt(-c/3);
      y1 := 2×rt×cf;
      y2 := -rt×(cf+ rt3×sf);
      y3 := -(d/y1)/y2;

      x1 := y1 - p;
      x2 := y2 - p;
      x3 := y3 - p;

      _i_f abs(x1)>abs(x2) _t_h_e_n Swap(x1,x2);
      _i_f abs(x2)>abs(x3) _t_h_e_n Swap(x2,x3);
      _i_f abs(x1)>abs(x2) _t_h_e_n Swap(x1,x2);

      w := x3;

      _i_f abs(x2) < 0.1×abs(x3) _t_h_e_n _g_o_t_o l70;
      _i_f abs(x1) < 0.1×abs(x2) _t_h_e_n x1 := -(r/x3)/x2;
      z[1,1] := x1;
      z[1,2] := 0;
      z[2,1] := x2;
      z[2,2] := 0;
      z[3,1] := x3;
      z[3,2] := 0;
      outputCode := 33;
      _g_o_t_o end;
      _c_o_m_m_e_n_t real and complex roots;
l40:  ra := CubeRoot(-0.5×d - abs(rq)×sign(d));
      rb := -c/(3×ra);
      t := ra + rb;
      w := -p;
      x := -p;
      _i_f abs(t) _< tol×abs(ra) _t_h_e_n _g_o_t_o  l41;
      w := t - p;
      x := -0.5×t - p;
      _i_f abs(x) _< tol×abs(p) _t_h_e_n x:=0;
l41:  t := abs(ra - rb);
      y := 0.5×rt3×t;

      _i_f t _< tol×abs(ra) _t_h_e_n _g_o_t_o l60;
      _i_f abs(x)<abs(y) _t_h_e_n _g_o_t_o l50;
      s := abs(x);

      t := y/x;
      _g_o_t_o l51;
l50:  s := abs(y);
      t := x/y;
l51:  _i_f s < 0.1×abs(w) _t_h_e_n _g_o_t_o l70;
      w1 := w/s;
      sum := 1 + t|∧2;
      _i_f w1|∧2 < 0.01×sum _t_h_e_n w := -((r/sum)/s)/s;
      z[1,1] := w;
      z[1,2] := 0;
      z[2,1] := x;
      z[2,2] := y;
      z[3,1] := x;
      z[3,2] := -y;
      outputCode := 32;
      _g_o_t_o end;
      _c_o_m_m_e_n_t At least two roots are equal;
l60:  _i_f abs(x)<abs(w) _t_h_e_n _g_o_t_o l61;
      _i_f abs(w)<0.1×abs(x) _t_h_e_n w := -(r/x)/x;
      z[1,1] := w;
      z[1,2] := 0;
      z[2,1] := x;
      z[2,2] := 0;
      z[3,1] := x;
      z[3,2] := 0;
      outputCode := 31;
      _g_o_t_o end;
l61:  _i_f abs(x) < 0.1×abs(w) _t_h_e_n _g_o_t_o l70;
      z[1,1] := x;
      z[1,2] := 0;
      z[2,1] := x;
      z[2,2] := 0;
      z[3,1] := w;
      z[3,2] := 0;
      outputCode := 31;
      _g_o_t_o end;
      _c_o_m_m_e_n_t
         Here w is much larger in magnitude than the other roots.
         As a result, the other roots may be exceedingly inaccurate
         because of roundoff errors. To deal with this, a quadratic
         is formed whose roots are the same as the smaller roots of
         the cubic.  This quadratic is then solved.
      ;

l70:  aq[1] := a[1];
      aq[2] := a[2] + a[1]/w;
      aq[3] := -a[4]×w;
      QuadraticRoots(aq, z);
      z[3,1] := w;
      z[3,2] := 0;
      outputCode := 31;
      _i_f z[1,2]=0 _t_h_e_n _g_o_t_o end;
      z[3,1] := z[2,1];
      z[3,2] := z[2,2];
      z[2,1] := z[1,1];
      z[2,2] := z[1,2];
      z[1,1] := w;
      z[1,2] := 0;
      outputCode := 32;
      _g_o_t_o end;
      _c_o_m_m_e_n_t Case when d = 0;
l110: z[1,1] := -p;
      z[1,2] := 0;
      w := sqrt(abs(c));

      _i_f c<0 _t_h_e_n _g_o_t_o l120;
      z[2,1] := -p;
      z[2,2] := w;
      z[3,1] := -p;
      z[3,2] := -w;
      outputCode := 32;
      _g_o_t_o end;
l120: _i_f p |= 0 _t_h_e_n _g_o_t_o l130;
      z[2,1] := w;
      z[2,2] := 0;
      z[3,1] := -w;
      z[3,2] := 0;
      outputCode := 31;
      _g_o_t_o end;
l130: x := -(p+abs(w)×sign(p));
      z[3,1] := x;
      z[3,2] := 0;
      t := 3×a[1]/(a[3]×x);
      _i_f abs(p)>abs(t) _t_h_e_n _g_o_t_o l131;
      z[2,1] := t;
      z[2,2] := 0;
      outputCode := 31;
      _g_o_t_o end;
l131: z[2,1] := z[1,1];
      z[2,2] := z[1,2];
      z[1,1] := t;
      z[1,2] := 0;
      outputCode := 31;
end:
   _e_n_d CubicRoots;
   _p_r_o_c_e_d_u_r_e QuarticRoots(a,z);
   _a_r_r_a_y a,z;
   _b_e_g_i_n
      _c_o_m_m_e_n_t
         PURPOSE - Compute the roots of the real polynomial
             a[1] + a[2]×z + ... + a[5]×z|∧4

         Note: It is assumed that a[5] is non-zero. No test is made here
      ;
      _r_e_a_l wr,wi,b,b2, c, d, e, h, p, q, r, t;
      _a_r_r_a_y temp[1:4];
      _r_e_a_l u, v, v1, v2, x, x1, x2, x3, y;
      DEBUGr(|<<a[1]|>, a[1]);
      _i_f a[1]=0 _t_h_e_n
      _b_e_g_i_n
         _a_r_r_a_y a3[1:4],z3[1:4,1:2];
         _i_n_t_e_g_e_r i;
         _c_o_m_m_e_n_t One root is obviously zero;
         z[1,1] := 0;
         z[1,2] := 0;
         _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l 4 _d_o a3[i]:=a[i+1];
         CubicRoots(a3,z3);
         _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l 4 _d_o
         _b_e_g_i_n
            z[i+1,1] := z3[i,1];
            z[i+1,2] := z3[i,2]
         _e_n_d;
         _g_o_t_o end
      _e_n_d;

      b := a[4]/(4×a[5]);
      c := a[3]/a[5];
      d := a[2]/a[5];
      e := a[1]/a[5];

      b2 := b|∧2;

      p := 0.5×(c - 6×b2);
      q := d - 2×b×(c - 4×b2);
      r := b2×(c - 3×b2) - b×d + e;

      _c_o_m_m_e_n_t

         Solve the resolvent cubic equation. The cubic has at least one
         nonnegative real root. If w1, w2, w3 are the roots of the cubic
         then the roots of the original equation are

            z = -b + csqrt(w1) + csqrt(w2) + csqrt(w3)

         where the signs of the square roots are chosen so
         that csqrt(w1) × csqrt(w2) × csqrt(w3) = -q/8
      ;

      temp[1] := -q×q/64;
      temp[2] := 0.25×(p×p - r);
      temp[3] := p;
      temp[4] := 1;
      DEBUGr(|<<temp[1]|>, temp[1]);
      DEBUGr(|<<temp[2]|>, temp[2]);
      DEBUGr(|<<temp[3]|>, temp[3]);
      DEBUGr(|<<temp[4]|>, temp[4]);
      CubicRoots(temp,z);
      DEBUGr(|<<z[1,1]|>, z[1,1]);
      DEBUGr(|<<z[1,2]|>, z[1,2]);
      DEBUGr(|<<z[2,1]|>, z[2,1]);
      DEBUGr(|<<z[2,2]|>, z[2,2]);
      DEBUGr(|<<z[3,1]|>, z[3,1]);
      DEBUGr(|<<z[3,2]|>, z[3,2]);
      _i_f z[2,2]|=0 _t_h_e_n _g_o_t_o l60;
      _c_o_m_m_e_n_t
         The resolvent cubic has only real roots
         Reorder the roots in increasing order
      ;
      x1 := z[1,1];
      x2 := z[2,1];
      x3 := z[3,1];
      _i_f x1>x2 _t_h_e_n Swap(x1,x2);
      _i_f x2>x3 _t_h_e_n Swap(x2,x3);
      _i_f x1>x2 _t_h_e_n Swap(x1,x2);

      u := 0;
      _i_f x3 > 0 _t_h_e_n u:=sqrt(x3);
      _i_f x2 _< 0 _t_h_e_n _g_o_t_o l41;
      _i_f x1 _> 0 _t_h_e_n _g_o_t_o l30;
      _i_f abs(x1) > x2 _t_h_e_n _g_o_t_o l40;
      x1 := 0;
l30:  x1 := sqrt(x1);
      x2 := sqrt(x2);
      _i_f q>0 _t_h_e_n x1 := -x1;
      temp[1] := (( x1 + x2) + u) - b;
      temp[2] := ((-x1 - x2) + u) - b;
      temp[3] := (( x1 - x2) - u) - b;
      temp[4] := ((-x1 + x2) - u) - b;
      DEBUGr(|<<temp[1] before sort|>, temp[1]);
      DEBUGr(|<<temp[2] before sort|>, temp[2]);
      DEBUGr(|<<temp[3] before sort|>, temp[3]);
      DEBUGr(|<<temp[4] before sort|>, temp[4]);
      SelectSort(4,temp);
      DEBUGr(|<<temp[1] after sort|>, temp[1]);

      DEBUGr(|<<temp[2] after sort|>, temp[2]);
      DEBUGr(|<<temp[3] after sort|>, temp[3]);
      DEBUGr(|<<temp[4] after sort|>, temp[4]);
      _i_f abs(temp[1]) _> 0.1×abs(temp[4]) _t_h_e_n _g_o_t_o l31;
      t := temp[2]×temp[3]×temp[4];
      _i_f t |= 0 _t_h_e_n temp[1] := e/t;
l31:  z[1,1] := temp[1];
      z[1,2] := 0;
      z[2,1] := temp[2];
      z[2,2] := 0;
      z[3,1] := temp[3];
      z[3,2] := 0;
      z[4,1] := temp[4];
      z[4,2] := 0;
      outputCode := 40;
      _g_o_t_o end;

l40:  v1 := sqrt(abs(x1));
      v2 := 0;
      _g_o_t_o l50;
l41:  v1 := sqrt(abs(x1));
      v2 := sqrt(abs(x2));
      _i_f q<0 _t_h_e_n u := -u;
l50:  x := -u - b;
      y := v1 - v2;
      z[1,1] := x;
      z[1,2] := y;
      z[2,1] := x;
      z[2,2] := -y;
      x := u - b;
      y := v1 + v2;
      z[3,1] := x;
      z[3,2] := y;
      z[4,1] := x;
      z[4,2] := -y;
      outputCode := 44;
      _g_o_t_o end;
      _c_o_m_m_e_n_t
         The resolvent cubic has complex roots
      ;
l60:  t := z[1,1];
      x := 0;
      _g_o_t_o _i_f t<0 _t_h_e_n l61 _e_l_s_e _i_f t=0 _t_h_e_n l70 _e_l_s_e l62;
l61:  h := abs(z[2,1]) + abs(z[2,2]);
      _i_f abs(t) _< h _t_h_e_n _g_o_t_o l70;
      _g_o_t_o l80;
l62:  x := sqrt(t);
      _i_f q>0 _t_h_e_n x := -x;
l70:  csqrt(z[2,1],z[2,2],wr,wi);
      u := 2×wr;
      v := 2×abs(wi);
      t := x - b;
      x1 := t + u;
      x2 := t - u;
      _i_f abs(x1) _< abs(x2) _t_h_e_n _g_o_t_o l71;
      t := x1;
      x1 := x2;
      x2 := t;
l71:  u := -x - b;
      h := u|∧2 + v|∧2;
      _i_f x1|∧2 < 0.01×MIN(x2|∧2,h) _t_h_e_n x1 := e/(x2×h);
      z[1,1] := x1;
      z[1,2] := 0;
      z[2,1] := x2;

      z[2,2] := 0;
      z[3,1] := u;
      z[3,2] := v;
      z[4,1] := u;
      z[4,2] := -v;
      outputCode := 42;
      _g_o_t_o end;
l80:  v := sqrt(abs(t));
      z[1,1] := z[3,1] := -b;
      z[1,2] := z[3,2] := v;
      z[2,1] := z[4,1] := -b;
      z[2,2] := z[4,2] := -v;
      outputCode := 44;
end:
   _e_n_d QuarticRoots;
   _p_r_o_c_e_d_u_r_e SelectSort(n,a);
   _v_a_l_u_e n;
   _i_n_t_e_g_e_r n;
   _a_r_r_a_y a;
   _b_e_g_i_n
      _i_n_t_e_g_e_r i,j;
      _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l n-1 _d_o
      _f_o_r j:=i+1 _s_t_e_p 1 _u_n_t_i_l n _d_o
      _i_f a[i]>a[j] _t_h_e_n Swap(a[i],a[j])
   _e_n_d SelectSort;
   _p_r_o_c_e_d_u_r_e DEBUGr(text, r);
   _v_a_l_u_e r;
   _s_t_r_i_n_g text;
   _r_e_a_l r;
   _b_e_g_i_n
      _i_f kbon  _t_h_e_n
      _b_e_g_i_n
         writecr;
         writetext(text);
         writetext(|<<:  |>);
         write(|<-d.dddddd_{10}-dd|>,r)
      _e_n_d
   _e_n_d;
   _i_n_t_e_g_e_r _p_r_o_c_e_d_u_r_e SolvePolynomial(quarticCoeff, cubicCoeff, quadraticCoeff,
       linearCoeff, constantCoeff, root1r, root1i, root2r, root2i, root3r, root3i,
       root4r, root4i);
   _v_a_l_u_e quarticCoeff, cubicCoeff, quadraticCoeff, linearCoeff, constantCoeff;
   _r_e_a_l quarticCoeff, cubicCoeff, quadraticCoeff, linearCoeff, constantCoeff,
      root1r, root1i, root2r, root2i, root3r, root3i, root4r, root4i;
   _b_e_g_i_n
      _a_r_r_a_y a[1:5],z[1:4,1:2];
      outputCode := -9999;
      a[1] := constantCoeff;
      a[2] := linearCoeff;
      a[3] := quadraticCoeff;
      a[4] := cubicCoeff;
      a[5] := quarticCoeff;
      DEBUGr(|<<quarticCoeff|>, quarticCoeff);
      _i_f quarticCoeff |= 0 _t_h_e_n
         QuarticRoots(a,z)
      _e_l_s_e _i_f cubicCoeff |= 0 _t_h_e_n
         CubicRoots(a,z)
      _e_l_s_e _i_f quadraticCoeff |= 0 _t_h_e_n
         QuadraticRoots(a,z)
      _e_l_s_e _i_f linearCoeff |= 0 _t_h_e_n
      _b_e_g_i_n
         z[1,1] := -constantCoeff/linearCoeff;
         z[1,2] := 0;
         outputCode := 1

      _e_n_d
      _e_l_s_e
         outputCode := 0;
      _i_f outputCode>0 _t_h_e_n
      _b_e_g_i_n
         root1r := z[1,1];
         root1i := z[1,2]
      _e_n_d;
      _i_f outputCode>1 _t_h_e_n
      _b_e_g_i_n
         root2r := z[2,1];
         root2i := z[2,2]
      _e_n_d;
      _i_f outputCode>23 _t_h_e_n
      _b_e_g_i_n
         root3r := z[3,1];
         root3i := z[3,2]
      _e_n_d;
      _i_f outputCode>39 _t_h_e_n
      _b_e_g_i_n
         root4r := z[4,1];
         root4i := z[4,2]
      _e_n_d;
      SolvePolynomial := outputCode
   _e_n_d SolvePolynomial;
   _r_e_a_l _p_r_o_c_e_d_u_r_e cabs(r,i);
   _v_a_l_u_e r,i;
   _r_e_a_l r,i;
   cabs:=sqrt(r|∧2+i|∧2);
   _p_r_o_c_e_d_u_r_e csqrt(xr,xi,zr,zi);
   _v_a_l_u_e xr,xi;
   _r_e_a_l xr,xi,zr,zi;
   _b_e_g_i_n
      _r_e_a_l r,sr,rzr;
      r := cabs(xr,xi);
      sr := sqrt(r);
      zr:=xr+r;
      zi:=xi;
      rzr := cabs(zr,zi);
      zr:=zr×sr/rzr;
      zi:=zi×sr/rzr;
   _e_n_d csqrt;
   _p_r_o_c_e_d_u_r_e Swap(a,b);
   _r_e_a_l a,b;
   _b_e_g_i_n
      _r_e_a_l t;
      t:=a;
      a:=b;
      b:=t
   _e_n_d Swap;
   _r_e_a_l _p_r_o_c_e_d_u_r_e MIN(a,b);
   _v_a_l_u_e a,b;
   _r_e_a_l a,b;
   MIN := _i_f a<b _t_h_e_n a _e_l_s_e b;
   _a_r_r_a_y coef[0:4],root[1:4,1:2];
   _i_n_t_e_g_e_r deg1, deg2, deg3, deg4;
   _r_e_a_l r1,r2,r3,r4;
   _a_r_r_a_y A1[0:1],A2[0:2],A3[0:3],A4[0:4];
   _a_r_r_a_y sort1,sort2[1:4];
   _i_n_t_e_g_e_r code,i;
   EPS := _r_e_a_l _1_0 996 _1_0 256 _1_0 0 _1_0 0;
   EPS4 := _r_e_a_l _1_0 998 _1_0 256 _1_0 0 _1_0 0;
   pi := 3.1415926536;
   oldrand := 100001;

   select(16);
again:
   coef[0] := RANDOM×10;
   coef[1] := -RANDOM;
   coef[2] := RANDOM×1_{10}-2;
   coef[3] := -RANDOM×1_{10}-4;
   coef[4] := RANDOM×1_{10}-6;
   r1 := RANDOM×10;
   r2 := RANDOM×10;
   r3 := RANDOM×10;
   r4 := RANDOM×10;
   sort1[1] := r1;
   sort1[2] := r2;
   sort1[3] := r3;
   sort1[4] := r4;
   A1[0] := r1;
   A1[1] := -1;
   deg1 := 1;
   Pmult(A2, deg2, A1, deg1, r2);
   Pmult(A3, deg3, A2, deg2, r3);
   Pmult(A4, deg4, A3, deg3, r4);
   writecr; write(|<-d.dddddd_{10}-dd|>,r1);
   writecr; write(|<-d.dddddd_{10}-dd|>,r2);
   writecr; write(|<-d.dddddd_{10}-dd|>,r3);
   writecr; write(|<-d.dddddd_{10}-dd|>,r4);

   _f_o_r i:=0 _s_t_e_p 1 _u_n_t_i_l 4 _d_o
   _b_e_g_i_n
      coef[i] := A4[i];
      writecr;
      write(|<dd|>, i);
      write(|<  -d.dddddd_{10}-dd|>, coef[i])
   _e_n_d;
   _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l 4 _d_o
   _b_e_g_i_n
      root[i,1] := 1234;
      root[i,2] := 5678;
   _e_n_d;
   code := SolvePolynomial(coef[4],coef[3],coef[2],coef[1],coef[0],
           root[1,1], root[1,2], root[2,1], root[2,2],
           root[3,1], root[3,2], root[4,1], root[4,2]);
   writecr;
   writetext(_i_f code=0 _t_h_e_n |<<degenerate equation|>
     _e_l_s_e _i_f code=1 _t_h_e_n |<<one real root|>
     _e_l_s_e _i_f code=21 _t_h_e_n |<<two identical real roots|>
     _e_l_s_e _i_f code=22 _t_h_e_n |<<two distinct real roots|>
     _e_l_s_e _i_f code=23 _t_h_e_n |<<two complex roots|>
     _e_l_s_e _i_f code=31 _t_h_e_n |<<multiple real roots|>
     _e_l_s_e _i_f code=32 _t_h_e_n |<<one real and two complex roots|>
     _e_l_s_e _i_f code=33 _t_h_e_n |<<three distinct real roots|>
     _e_l_s_e _i_f code=40 _t_h_e_n |<<four real roots|>
     _e_l_s_e _i_f code=42 _t_h_e_n |<<two real and two complex roots|>
     _e_l_s_e _i_f code=44 _t_h_e_n |<<four complex roots|>
     _e_l_s_e _i_f code=-9999 _t_h_e_n |<<unassigned code|>
     _e_l_s_e |<<unknown code|>);
   _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l 4 _d_o sort2[i] := root[i,1];
   SelectSort(4,sort1);
   SelectSort(4,sort2);
   _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l 4 _d_o
   _b_e_g_i_n
      writecr;
      write(|<  -d.dddddd_{10}-dd|>, sort1[i], sort2[i], sort1[i]-sort2[i]);
   _e_n_d;
   _f_o_r i:=0 _s_t_e_p 1 _u_n_t_i_l 3 _d_o

   _b_e_g_i_n
      coef[i] := A4[i+1]×(i+1);
      writecr;
      write(|<dd|>, i);
      write(|<  -d.dddddd_{10}-dd|>, coef[i])
   _e_n_d;
   _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l 4 _d_o
   _b_e_g_i_n
      root[i,1] := 1234;
      root[i,2] := 5678;
   _e_n_d;
   code := SolvePolynomial(0,coef[3],coef[2],coef[1],coef[0],
           root[1,1], root[1,2], root[2,1], root[2,2],
           root[3,1], root[3,2], root[4,1], root[4,2]);
   writecr;
   writetext(_i_f code=0 _t_h_e_n |<<degenerate equation|>
     _e_l_s_e _i_f code=1 _t_h_e_n |<<one real root|>
     _e_l_s_e _i_f code=21 _t_h_e_n |<<two identical real roots|>
     _e_l_s_e _i_f code=22 _t_h_e_n |<<two distinct real roots|>
     _e_l_s_e _i_f code=23 _t_h_e_n |<<two complex roots|>
     _e_l_s_e _i_f code=31 _t_h_e_n |<<multiple real roots|>
     _e_l_s_e _i_f code=32 _t_h_e_n |<<one real and two complex roots|>
     _e_l_s_e _i_f code=33 _t_h_e_n |<<three distinct real roots|>
     _e_l_s_e _i_f code=40 _t_h_e_n |<<four real roots|>
     _e_l_s_e _i_f code=42 _t_h_e_n |<<two real and two complex roots|>
     _e_l_s_e _i_f code=44 _t_h_e_n |<<four complex roots|>
     _e_l_s_e _i_f code=-9999 _t_h_e_n |<<unassigned code|>
     _e_l_s_e |<<unknown code|>);
   _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l 4 _d_o
   _b_e_g_i_n
      writecr;
      write(|<  -d.dddddd_{10}-dd|>, root[i,1], root[i,2]);
   _e_n_d;
   _g_o_t_o again;
bad:
_e_n_d;
t<